```
DIV_3216U_ATFN              ;unsigned 32/16 bits values division

;To call the routine:
;dividend in DEND_3:0 (max val H'FFFE 0001' = H'FFFF' * H'FFFF' =4.294.836.225)
;divisor in DSOR_H:L (max value H'FFFF' =65.535).
;User to ensure being within range or if a division by zero is going to happen.
;User to ensure the quotient is limited to 16-bits.

;The routine gives:
;quotient in DEND_1:0 (16 bits)
;reminder in DEND_3:2 (16 bits)

;Result of DEND_3:DEND_0 / DSOR_1:DSOR_0 => DEND_1:DEND_0 (quotient)
;Remainder in DEND_3:DEND_2

    LOADREG CNTR_SHIFTS,16  ;shiftings counter

DIV_3216U_ATFN_LOOP
    BCF STATUS,C            ;we ensure that b0
    RLCF DEND_0,F           ;of DEND_0 is =0
    RLCF DEND_1,F           ;after the shifting
    RLCF DEND_2,F           ;to the left of the
    RLCF DEND_3,F           ;whole dividend

;We test C. If =1, shifting o'flow!! Substract DSOR_H:L from DEND_3:2
    BC DIV_3216U_ATFN_SUBST_DSOR

    MOVF DSOR_H,W           ;compare value of DSOR_H
    SUBWF DEND_3,W          ;against DEND_3

;We test Z. If =0, DSOR_H != DEND_3. Chech if  DSOR_H < DEND_3
    BNZ DIV_3216U_CHKIF_DSORH_LOW_DEND3

    MOVF DSOR_L,W           ;DSOR_H = DEND_3; now compare
    SUBWF DEND_2,W          ;value of DSOR_L against DEND_2

;We test C. If =1, DSOR_L < DEND_2. Substract DSOR_H:L from DEND_3:2
    BC DIV_3216U_ATFN_SUBST_DSOR

DIV_3216U_ATFN_DECR_CNTR_SH
    DECFSZ CNTR_SHIFTS,F
        BRA DIV_3216U_ATFN_LOOP
    RETURN                  ;job done!

DIV_3216U_ATFN_SUBST_DSOR   ;substract DSOR from MSBs of dividend
    MOVF DSOR_L,W           ;DEND_2 = DEND_2 - DSOR_L
    SUBWF DEND_2,F          ;Borrow not used (LSB)
```

```
    MOVF DSOR_H,W           ;DEND_3 = DEND_3 - DSOR_H
    SUBWFB DEND_3,F         ;Borrow used

    BSF DEND_0,0            ;flag "divisor substracted from dividend"
    BRA DIV_3216U_ATFN_DECR_CNTR_SH ;go for next shift

DIV_3216U_CHKIF_DSORH_LOW_DEND3
    BNC DIV_3216U_ATFN_DECR_CNTR_SH ;C =0, can't substract (DSOR_H > DEND_3)
    BRA DIV_3216U_ATFN_SUBST_DSOR   ;C =1, can substract (DSOR_H < DEND_3)
;-----------------------------------------------------------------------------
```